
 

 

Chapter 11. Building Event Networks 
In the preceding chapters, we’ve focused on rules that respond to events. 
But there’s another way that rules and events interact: rules can raise 
events. The power of KRL expands significantly when rules raise events 
because rules can be used to build event networks. We call events that are 
raised by a rule “explicit events.”  
One of the key design goals for rule languages is to create rulesets that are 
loosely coupled. Rules should add behavior in an accretive manner so that 
rules can be added or deleted without affecting the behavior of other rules.  

This chapter will explore explicit events and how they can be used to 
create event networks. We will see that explicit events provide an 
important means of abstraction in KRL through event synthesis and rule 
chaining. A number of event intermediary patterns present themselves 
once we can raise explicit events.  

Understanding Event Types 
There are two ways to create an explicit event. Explicit events can be 
raised in the rule postlude using the raise statement or they can be 
raised as an action using the raise_event() action.  

The diagram in Figure 1 shows the difference between these two kinds of 
explicit events. An explicit event in the postlude handles the event in the 
same processing episode as the original event and results in a possibly 
increased set of directives being sent back to the endpoint. The explicit 
event causes further processing to happen on the rules engine without 
further involvement by the endpoint.  

Figure 1. Explicit Event Types 
In contrast, the raise_event() action sends instructions back to the 
endpoint that cause it to raise another event. The disadvantage is that there 
is a second round trip from the endpoint to KRE made in response to the 
action. The advantage is that information on the endpoint can be sent 
along with the new event and is thus available for processing on the rules 
engine.  

When an event is raised in the postlude, the event will have the event 
domain explicit, so you would use an eventex like the following to 
write a rule that is selected when this event is raised: 

select when explicit <event_type> 

When an event is raised by the endpoint in response to a 
raise_event() action, the new event will have the event domain of 
the endpoint raising the event. The following eventex would select a rule 
based on an explicit event raised from the Web endpoint: 

select when web <event_type> 

Raising Explicit Events in the Postlude 

Explicit events are raised in the rule postlude with a raise statement:  

raise explicit event foo for “a16x48” 

    with a = "hello" 

     and b = 4 + x;  

The for clause is optional. If it’s missing, the event will be raised within 
the current ruleset. The for clause is followed by an expression that is 
evaluated to determine which rulesets should see the event. The result of 
the expression should be either a single string or an array of strings. The 
strings represent ruleset IDs.   

The with clause allows the developer to add event parameters to the 
explicit event. The right-hand side of the individual bindings in the with 
clause can be any KRL expression.  

Like any other postlude statement, explicit events can be guarded:  

raise explicit event foo 

      with a = "hello" 

       and b = 4 + x 

   if (flipper == "two");  

The event in the preceding example will only be raised if the variable 
flipper has the value “two.”  

Explicit events allow KRL programmers to chain rules together. Rule 
chaining is good for modularization, error handling, preprocessing, and 
abstraction as we’ll show in the following sections.  

Automatically Raising Events 
Sometimes, it’s useful to explicitly raise an event to handle responses from 
an action1. For example, when you use http:post() as an action, you 
might want to respond to status codes indicating an error. Explicit events 
can be used to cause another rule to handle that response.  
While we can simply use the raise statement in the postlude as we just 
discussed, this happens often enough for certain actions that they support 
shorthand for raising explicit events from the action.  

Certain actions automatically raise events using the optional autoraise 
parameter2. The autoraise parameter is given with a string as its value. 

When the explicit event is automatically raised, one of the event 
parameters will be label and it will have the value of the autoraise 
string. 

Each action defines its own event domain and type. For the 
http:post() action the event that is raised will have event domain 
http and event type of post. Similarly, each action will include relevant 
data from the action.  The response values given above are sent as event 
parameters and can thus be checked as part of the event selection. The 
http:post() action includes response information from the POST.  

The following simple example shows a rule that has an http:post() 
action with an event autoraise: 

rule r1 is active { 

  select when pageview "/archives/(\d+)/" setting(year) 

  http:post("http://www.example.com/go") 

      with params = {"answer": "x"} and 

           autoraise = "ex"; 

} 

This is roughly equivalent to the following rule: 

rule r1 is active { 

  select when pageview "/archives/(\d+)/" setting(year) 

  http:post("http://www.example.com/go") setting (resp) 

      with params = {"answer": "x}; 

  always { 

    raise explicit event post with resp 

  } 

} 

The only difference is that the autoraise would create an event with 
the event domain http while the raise statement would create an event 
with the event type explicit.  

Assuming we automatically raised an event in the first rule shown above, 
we could chain additional rules for subsequent processing of the response. 
The following two rules check the status code of the response and present 
a notification of the result:  

rule r2 is active { 

  select when http post label re#ex# status_code re#(2\d\d)# 

     setting (status) 

  notify("Status", "Success! The status is " + status); 

}  

rule r3 is active { 

  select when http post label re#ex# status_code re#([45]\d\d)#  

    setting (status) 

  fired { 

    log <<Error: #{status}: #{event:attr(“status_line”)}>> 

    last; 

  } 

} 

The second rule fires when the status code in the response indicates an 
error, logs the error, and uses the last control statement in the postlude 
to stop subsequent processing of rules in the ruleset.  
We can also process responses based on other event attributes like the 
content type. This example rule shows the content of the response if its 
content type is “text.” 

rule r4 is active { 

  select when http post label re#ex# 

  if(event:attr("content_type") like "^text/") then  

    notify("Page says...", event:attr("content")); 

} 

Rule chaining from an autoraise on an http:post() action 
provides a convenient, event-driven way of dealing with the results of an 
action.  

The Explicit Event Action 
The Web endpoint supports a raise_event() action that causes the 
runtime to raise an event.  The raise_event() action causes control to 
be passed back to the rules engine from the Web endpoint for further 
processing3.  
The raise_event() action takes a string argument that declares the 
event name and two optional parameters: app_id and parameters.  
The app_id identifies the ruleset for which the event will be raised. If 

the app_id is missing, the event is raised to the current ruleset. 
Parameters is a map of name value pairs.  

raise_event(<event_type>)  

    with app_id = <rid> and  

         parameters = {“name0”: value0, 

                       ... 

                       “namen”: valuen 

                      };  

We can see how this works with an example. Suppose ruleset A has the 
following rule: 

rule Raise_Event_Action { 

  select when pageview ".*" 

  { 

    notify("Kynetx Event Walkabout", 

           "Raise_Event_To_Remote_Ruleset"); 

    raise_event("event_remote_ruleset")  

      with app_id = “B”; 

  } 

} 

And ruleset B contains the following code: 

rule Catch_Remote_Event { 

  select when web event_remote_ruleset 

  notify("Catch_Remote_Event",  

         “Received from remote Ruleset!") 

} 

When Raise_Event_Action is selected, the notification 

boxes shown in  
Figure 2 will show up in the user’s browser. 

 

Figure 2. Simple raise_event() action interplay 
We can make this simple example more interesting by making ruleset B 
callback to the ruleset that raises the event it processes. First, add the 
following rule to ruleset A: 

rule Catch_Event_Callback { 

  select when web event_callback 

  notify("Catch_Event_Callback",  

         "Received Callback Event!")  

} 

We also add some parameters to the raise_event() action in the 
Raise_Event_Action rule in ruleset A:  

rule Raise_Event_Action { 

  select when pageview ".*" 

  { 

    notify("Kynetx Event Walkabout", 

           "Raise_Event_To_Remote_Ruleset"); 

    raise_event("event_remote_ruleset")  

      with app_id = “B”  

       and parameters = { 

                "callback_rid" : “A”, 

                "callback_evt" : "event_callback" 

                } 

  } 

} 

Now we change the rule in ruleset B to look for and process the callback: 

rule Catch_Remote_Event { 

  select when web event_remote_ruleset 

  pre { 

    callback_rid = event:attr("callback_rid"); 

    callback_evt = event:attr("callback_evt"); 

  } 

  { 

    notify("Catch_Remote_Callback",  

           "Received from remote Ruleset!"); 

    notify("Now Raise Callback",  

           "rid: #{callback_rid} name: #{callback_evt});  

    raise_event(event_callback)  

      with app_id = callback_rid; 

  } 

} 

Note that Catch_Remote_Event uses the event type and 
ruleset ID that are set in Raise_Event_Action to 

determine which event to raise and specify the ruleset 
that will process it. The Catch_Event_Callback rule we 

added to ruleset A catches the callback causing the 

notification boxes in  
Figure 3 to appear.  

 

Figure 3. Explicit event action with callback 

Explicit Events as Rule Abstractions 

In Chapter 7, we created a ruleset that greeted people by name if an entity 
variable with their name was present. Otherwise, the ruleset placed a form 
on a page asking for their name and then responded by showing the name 
submitted in place of the form. Recall that the rule that responded to the 
submit event looked like this:  

rule respond_submit { 

   select when web submit "#my_form" 

   pre { 

     name = event:attr("first")+“ “+event:attr("last")}; 

   } 

   replace_inner("#my_div", "Hello #{name}"); 

   fired { 

      mark ent:name with name; 

   } 

 }    

The rule that used the data and replaced the form with the name looked 
like this: 

rule replace_with_name { 

  select when web pageview “.*” 

  pre { 

    name = current ent:name; 

  } 

  replace_inner("#my_div", "Hello #{name}"); 

}  

The problem with this design is that we’re using replace_inner() in 
two different places to paint the greeting on the page. Experienced 
programmers try to avoid doing the same thing in two places because it 
leads to maintenance problems. For example, if we change the greeting to 
“Howdy #{name}!” we have to remember to change it in two places. 

The answer to problems like this is almost always some kind of 
abstraction. We saw one possible solution, changing the eventex of the 
second rule, in Chapter 7.  Since Chapter 7, we’ve learned about user-
defined actions, which is another important kind of abstraction in KRL. So 
we could define an action called send_greeting() that abstracts the 
greeting like so: 

send_greating = defaction(name) { 

    replace_inner("#my_div", "Hello #{name}") 

  } 

Replacing the action in both rules with send_greeting() would 
abstract away the details of the greeting and put them in one place. Now 
we have one place to change the greeting, but we are still taking the same 
action in two places.  

Another way to deal with this problem is to use an explicit event to chain 
the rules, letting just one rule send the greeting. The first rule would now 
look like this: 

rule respond_submit { 

   select when web submit "#my_form" 

   pre { 

     name = event:attr("first")+“ “+event:attr("last")}; 

   } 

   always { 

      mark ent:name with name; 

      raise explicit event send_greeting 

   } 

 }    

The only purpose of this rule is to record the response to the Web form 
submission.  
The second rule is changed to select when there is an appropriate 
pageview event or an explicit event of the right type: 

rule replace_with_name { 

  select when web pageview “.*” or explicit send_greeting 

  pre { 

    name = current ent:name; 

  } 

  replace_inner("#my_div", "Hello #{name}"); 

}  

With this change we have separated the functionality of each rule so that 
each does one thing. The first rule stores the value of the submission. The 
second rule sends the greeting. This creates a clean separation of concerns 
among rules in the ruleset. The rules create the desired functionality 
accretively. This is a key design goal in rule-based programming.  

Explicit Events for Error Handling 
Now that we have an understanding of explicit events in KRL, we’re 
prepared to explore the way that KRL provides for error handling. Events 
are a natural way to handle errors.  

Raising Errors 
KRE will raise errors for various system level errors that happen during 
the execution of a KRL ruleset. For example, KRL will raise an error 
when a rule attempts to take an undefined action or for a type mismatch on 
an operator (e.g. applying the length() operator to an integer)4.  

When KRE raises an error event it uses the event domain system and the 
event type error. The following event attributes are attached to the 
event: 

• level—the level of the event: error, warn, info, or debug. 
Processing continues for all levels except error where execution 
is terminated. 

• msg—a string giving details about the error 

• rid—the ruleset ID of the offending ruleset 

• rule_name—the name of the offending rule 

• genus—a token from the top level of the event taxonomy 
described below indicating the first level classification of the error. 

• species—a token from the second level of the event taxonomy 
indicating the classification of the error within the genus 

In addition to system errors, KRL programs can also raise errors explicitly 
in the rule postlude. The syntax of an explicit error statement is: 

error <level> <expr>  

In this statement <level> is one of error, warn, info or debug 
and <expr> is any valid KRL expression that results in a string (or 
something that can be cast as a string such as a number). 
The following example would raise an event with domain system and 
type error with level info and a message with the value of a variable 
named query if the rule fired: 

fired {    

  error info "query:"+query  

}  

The following would only raise the event if the query variable is empty: 

fired { 

   error info "Empty query" if(query like "^$") 

}  

Explicit errors set the rid and rule_name attributes from the current 
ruleset ID and rule name.  The genus is set to “user.”  

Handling Errors 
Handling errors is as easy as creating a rule with the right select 
statement. For example, the following rule will use the send_error() 
action from a ErrorStack module to record an error using the ErrorStack 
service: 

rule process_error { 

  select when system error 

  pre{ 

    genus = event:attr("genus"); 

    species = event:attr("species"); 

  } 

  es:send_error("(#{genus}:#{species}) " + event:param("msg")) 

       with rule_name = event:param("rule_name") 

        and rid = event:param("rid"); 

} 

Like any other event, if an error event is raised and no rule is selected for 
it, nothing happens.  
Because developers will often want to process all errors from several 
rulesets in a consistent way, KRL provides a way of routing error events 
from one ruleset to another. In the meta section of a ruleset, developers 
can declare another ruleset that is the designated error handler. 

For example, if the preceding process_error rule were defined inside 
ruleset a16x88, then the following declaration would route error events 
from the current ruleset to a16x88 for processing: 

meta {   

  errors to a16x88 

} 

Raising events for errors and then handling them with rulesets gives 
developers a great deal of flexibility in dealing with exceptional situations 
in their code.  

Events and Loose Coupling 
One of the dangers of explicit events is that we can use explicit events to 
introduce a form of tightly coupled “rule calling” in our rulesets. The 
example with rule chaining in the preceding section is a good example. 
The respond_submit rule is not really raising a general-purpose event 
of broad interest, but rather a narrowly focused event that is designed for a 
single purpose: forcing the replace_with_name rule to fire.  

Rule chaining as we introduce in the last section isn’t likely to cause many 
problems in designing a KRL program, but you should be aware that 
creating lots of narrowly focused, specific events as a way of explicitly 
“calling” certain rules runs the danger of introducing tight coupling into 
your applications. As we’ll see in coming chapters, broad-interest events 
can lead to new opportunities to expand the application easily later on.   

Remember that the purpose of raising an event is to send a notification, 
not call a rule. Even the event name we chose, send_greeting, belies 
our intention of making a request, rather than making a notification. In 
fact, if you find yourself thinking about “calling” rules then you’re 
probably doing it wrong.  
Instead, suppose the raise statement had read instead: 

raise explicit event form_results_posted 

This would, more readily, lead to thinking that allows multiple rules to 
listen for and respond to the event. We might, for example, add a rule that 
uses the results for a database lookup once they’re posted. With an event 
named send_greeting, we’re less likely to see that possibility since 
we view it as a request to do a certain thing.  

Event Intermediaries 
One of the key ideas of distributed systems is using intermediaries to 
filter, augment, route, and process data in between its origin and ultimate 
destination. Intermediaries can reduce the volume of data that the 
processor handles as well as improving its quality. Intermediaries make 
use of the age-old computer idea: that a level of indirection can cure many 
ills. Intermediaries go by different names in different systems: 
middleware, routers, proxies, gateways, and so on.  

 

Figure 4. Event intermediaries sit between the endpoint 
and the rules that process the events 

As shown in 

 
Figure 4, event intermediaries sit in between the endpoint and the rules 
that ultimately service them. They provide an improved stream of events 
for the rules and make their implementation simpler by pre-processing the 
events.  
The general idea behind all event intermediaries is event synthesis: 
creating new, meaningful events from the originals. Synthesized events 
appear to be simple events, but are the result of a planned computational 

process that used complex streams of events as input, processed them in 
some way, and creates a new event that contains the relevant data from the 
originals.  
In KRL, event intermediaries are rules. Intermediary rules are just like 
normal KRL rules. They can be in the same ruleset as the other rules in an 
application or kept separate. One thing that most event intermediary rules 
have in common is that they take no action. The rule consists of an event 
expression, some data manipulation in the prelude, and a postlude that 
raises the new event.  
The next section will explore some common intermediary patterns. 

Event Intermediary Patterns 
Intermediaries can take many forms, so we will explore them by looking 
at some common intermediary patterns.  

Event Logging 
One of the simplest intermediary patterns is the event logging pattern. The 
intermediary rule looks for the expected event scenario, calls a logging 
statement (either using the built-in log command in KRL or by making an 
HTTP post) and then passes the event on using an explicit event.  

The following rule illustrates this by using http:post() to create a 
log:  

rule logger_rule is active { 

   select when phone outboundconnected 

   http:post("http://example.com/mylogger.cgi") with 

      with number = event:attr("phonenumber") 

   always { 

     raise explicit event outboundconnected with 

        phonenumnber = event:attr("phonenumber") and 

        time = event:attr("time"); 

   } 

} 

rule use_phone { 

  select when explicit outboundconnected  

  ... 

} 

In this example, the rule is logging the event and some data from it before 
passing the event on as an explicit event. Event logging might be used for 
debugging, billing, ruleset analytics, and so on.  

Abstract Event Expressions 
Because of the nature of rule languages, you may often write several rules 
that have the same eventex. This goes against the grain of programmers of 
traditional programming languages where repeating yourself is not only 
wasteful but leads to code maintenance problems.  
The right response is to abstract the portions of those multiple rules that 
are repetitive and that are apt to be changed frequently. We’ve seen how 
functions, user-defined actions, and modules can help with that. But none 
of these can help with a repeated complex eventex. Explicit events solve 
that problem. In addition to maintainability, using explicit events to 
abstract a complex eventex by giving it a name can facilitate program 
readability  

In the following example, we name a complex eventex. The first rule 
contains the complex eventex and names it called_first: 

rule called_first is active { 

  select when phone outboundconnected 

       before email received to re#@apple.com# 

  always { 

    raise explicit event called_first  

      with msg = event:param("msg"); 

  } 

} 

rule use_called_first_1 is active { 

  select when explicit called_first 

  ... 

} 

rule use_called_first_2 is active { 

  select when explicit called_first 

  ... 

} 

Notice that the first rule raises an explicit event with the name 
called_first whenever it sees a particular event pattern. Two later 
rules use the called_first event. If the complex event expression is 
changed or updated the two rules will both respond appropriately. When 
the event expression is used in this manner, we call called_first an 
abstract or named event expression.  

Event Preprocessing 
Sometimes event parameters need to be preprocessed before they are used. 
Based on the results of the preprocessing, you may want to do different 
things. Preprocessing is a way of enriching the event by using the event 
parameters in some way; for example, looking up relevant data from 
online data sources and then sending along the results.  
Preprocessing is an important form of event abstraction because when we 
enrich an event or preprocess the event parameters we avoid doing the 
same calculations multiple times in other rules.  

The following rule pre-processes an email event to look up data from an 
online data source, making the data in the message body more relevant: 

rule pinentered is active { 

  select when email received 

  pre { 

    msg = event:attr("msg"); 

    from = event:attr("from"); 

    item = datasource:pds({"key":from}); 

    relevant_data = msg.query("li[type=#{item}]"); 

  } 

  always { 

    raise explicit event mail_received with 

      from = event:attr("from") and 

      to = event:attr("to") and 

      msg = relevant_data 

  } 

} 

In this example the message from an email that’s been received is 
preprocessed using the query operator to retrieve just those portions of the 
message that are HTML <li> elements with an attribute named type 
equal to a value that is retrieved from a data source named pds using 
“From” address of the email.  

Often it’s helpful to perform a complex mapping step once. Using explicit 
events we can put the preprocessing in a single place where it can be more 
easily maintained and tested.  

Event Stream Splitting 
Related to the idea of event preprocessing is the notion of event ‘stream 
splitting’. The previous example shows how to perform event parameter 
preprocessing. We can use the event parameters to split the event stream 
and send it in two different directions. Often preprocessing will be done in 
support of splitting the event stream.  

The following rule preprocesses the event data and then uses a rule 
condition to raise one of two possible explicit events depending on the 
result of the condition:  

rule pinentered is active { 

  select when webhook pinentered 

  pre { 

    pinattempt = event:param("Digits"); 

    phone = datasource:pds({"key":"phone"}); 

    pin = phone.pick("$..value.pin"); 

  } 

  if pinattempt == pin then 

    noop(); 

  fired { 

    raise explicit event correct_pin 

  } else { 

    raise explicit event bad_pin 

  } 

} 

In this example the data in the event attribute Digits is compared with 
data retrieved from another data source (datasource:pds). If they’re 

equal, the rule raises the explicit event correct_pin, otherwise the rule 
raises the explicit event bad_pin. subsequent rules continues processing 
as necessary. In this case none of the original event’s data is passed on 
with the new events, but that need not be the case.  

App Controller Ruleset 
Complex apps will often be based on more than one ruleset.  We’ve built 
some that use a dozen or so and I expect to see apps that use many more 
than that. One of the problems when building complex applications that 
comprise multiple rulesets, is keeping track of the control points in the 
app—which events are causing which behavior. Developers often want a 
single place in the code where they manage control flow.  
Using the patterns outlined above, you can create a controller ruleset that 
is the main entry point for the app and controls the rules that get executed 
in other rulesets. Here are a few of the advantages of using a controller 
ruleset in your app:  

• routing - Each complex event pattern that the app responds to is 
represented in the controller ruleset. Each of these event patterns 
raises an explicit event to which the other rules in the app respond 
(event abstraction).  

• authentication – When you have an app that needs to be 
authenticated, you typically will also need a single place to control 
the authentication. An event controller solves this problem by 
being the one point of control and thus serving as the place where 
authentication can be controlled as well.  

• normalization - preprocessing event parameters in the controller 
app provides a normalized version of data and can serve to insulate 
the rest of the app from changes in outside event sources and 
endpoints. 

Complex Event Scenarios 
In Chapter 3 we explored KRL event expressions and the multitude of 
scenarios that can be described with them. Because eventexs are roughly 
equivalent in computing power to regular expressions, there will always 
be scenarios that are too complex to express using eventexs alone. Explicit 
events allow rules to be chained together to create scenario detectors that 
are as complex as they need to be.   

The answer to this problem is to create an intermediate rule that 
implements a recognition engine for the scenario.  To understand this, 
consider this eventex example that we introduced in Chapter 3. The 
eventex is designed to select when an RSS feed contains a story that: 
includes a stock ticker symbol, and the price of that same stock goes up by 
more than 2% within 10 minutes: 

select when rss item content re#Stock Symbol: (\w+)#  

              setting (symbol) 

     before stocks price_rise ticker eq symbol && percent > 2 

  within 10 minutes 

Suppose however that we want to be more sophisticated and vary the 
percentage gain and the time frame according to the levels shown in 
Error! Reference source not found.. There’s no good way to do that 
from a single eventex.  

Table 1. Urgency levels for stock price rises 
Percent Time (min) Level 
2 10 Urgent 
1.5 30 High 
1 60 Medium 
0.5 360 Low 
 

We can, however, write rules that abstract the percent gain and time as 
shown in the table and raise another event with the result. The first rule 
merely captures the timestamp of the stock mention and passes it on: 

rule stock_mention { 

  select when rss item content #Stock Symbol: (\w+)#  

             setting (sym)  

  always { 

    raise explicit event “stock_rise” with  

      rise_timestamp = event:attr(“timestamp”) and 

      symbol = sym; 

}  

The second rule does all the work, calculating the elapsed time and then 
setting the urgency according to Error! Reference source not found.. 
The result is an explicit event that has parameters for the relevant 
information. Other rules can then key off this explicit event. 

rule stock_rise { 

   select when explicit stock_rise symbol re#.*# setting(symbol) 

        before stocks price_rise ticker eq symbol && percent > 0.5 

       within 360 minutes 

  pre { 

   elapsed = time:diff(event:attr(“rise_timestamp”), 

                       time:now(), ”minutes”); 

   percent = event:param(“percent”); 

   urgency = (elapsed < 10 && percent > 2)   => “urgent” | 

             (elapsed < 30 && percent > 1.5) => “high” | 

             (elapsed < 60 && percent > 1)   => “medium” | 

                                                “low”; 

  } 

  always { 

   raise explicit event stock_rise with  

     symbol = symbol and 

     percent = percent and 

     urgency = urgency and 

     elapsed = elapsed 

  } 

} 

Together these two rules serve to create a detector for a complex event 
scenario that is difficult, if not impossible, to express in a single eventex.  

Explicit Events 
Along with functions, user-defined actions, and parameterized modules, 
explicit events represent one of the key abstraction mechanisms in KRL.  
Explicit events allow rule chaining and event synthesis, both a key to 
abstraction.  
Explicit events open up the use of event intermediaries in KRL and 
significantly expand the viability of complex apps built from multiple 

rulesets. Explicit intermediary rulesets like an app controller greatly 
reduce the cognitive complexity of large applications.  

Endnotes 
                                                             
1 This is different from the raise_event() action that we’ll discuss in 
the next section.  
2 Currently only the post action the http module supports 
autoraise.  
3 While only the Web endpoint currently supports the raise_event() 
action, future endpoints in other domains will also support it since it 
provides important control flow. 
4 For a complete list of the error types raised by the system, see the KRL 
documentation.  


